
I know that Lertap 5 users frequently feed scanner-produced data into Excel, and from there proceed to set up a workbook so that it’s ready for Lertap.
This document looks at a very practical example of scanner-based work on-going at a major testing centre at an eastern Australian university.  It walks through the process of modifying scanner data for Lertap, and shows how to get Lertap to produce reports and graphs in a turn-key manner, that is, without actually having to click on Lertap toolbar options.
To begin, have a look at a sample Excel workbook, “ScannerEjem​ploDP2”, produced by a scanner feeding its results into a computer equipped with the Microsoft Jet database engine:

As you can see, people at the testing centre have trained their scanner / computer combo so that it creates an Excel workbook with two worksheets.  One of the worksheets, RawData, holds student data and the item responses, with the item responses starting in column 9 and extending, in this case, through column 88.  These results are from an 80-item multiple-choice test, with each item using {A, B, C, D} as response options.
The Answers worksheet has the item keys in column 3, rows 2 through 81.

The format seen in these two worksheets is constant.  Item responses always start in column 9 of the second row of RawData; item keys always start in row 2 of the third column of the Answers worksheet.  Student ID numbers are always found in column 2 of the RawData worksheet.  This is true no matter what test has been run through the scanner; some of the testing centre’s standard achievement tests have less than 30 items, while some have up to 200.
Thus, the width of the RawData worksheet, and the length of the Answers worksheet, depend only on the number of items processed.  The RawData worksheet will always have its first eight columns for student and administrative data, followed by “Nits” columns with item responses, where Nits equals the number of test items.

The length of the Answers worksheet, that is, the number of rows, will always be Nits + 1.
Making the workbook Lertap-ready

If we opened Lertap at this point, bringing in its toolbar, would we expect the options under the Run menu, such as “Interpret CCs lines” and “Elmillon item analysis” to run with the ScannerEjem​ploDP2 workbook?

Of course not.
Lertap expects to have a workbook with item responses in a worksheet named Data, and also requires another worksheet, the all-important CCs worksheet with Lertap control information.

To this point the ScannerEjemploDP2 workbook is lacking these worksheets.  I have to create them.

I’ll begin by making a copy of the RawData worksheet.  This I do by right-clicking on the worksheet’s tab, and then clicking on Move or Copy ….

This leads to the following little dialog box:

Notice how I have ticked the little box in front of the Create a copy option?  This is muy importante (very important).  I end up with a new worksheet in my workbook:
[image: image5.png]E3 Microsoft Excel - ScannesEjemploDP2.xls

@) o & Yen Iset Foms Iob DSa Wndow DoumentsToGo

Help POF Create! -8 x

1 2 4 9 10 11 12 | 13
Batch Student ID Sumname Q1 Q2
25242 Buxwwwkx  Binhex
25242 3wk Cimeok

25242 Buwwwwxx  Demsat

25242 e Haines
v\ RawData (2) { Rewbata £ Answer: |« ||

1
2
3
4 |25242 3w Conetab
5
6
4

i«




Next, I rename RawData (2) to Data, then I insert a new row at the top, and type some sort of description into it:

I also changed the heading on column 2.  It was Student ID; now it’s become ID number.  Because why?, as my dear late Jakarta colleague Alex Putinella would undoubtedly ask.

Do you know the answer?  If I want the Lertap Scores report to include student IDs, the title of the respective column has to begin with “ID”.  (This is covered in Chapter 3 of the manual.  Note that the manual says that the ID has to reside in either the first column, or the second—this is no longer true; the ID may now be in any column of the Data worksheet.)
So, am I ready to Lertap?

Nope.  I need a CCs worksheet.  Here’s a suitable one:

So, am I finally set to rock and roll?  Almost.  I just need Lertap’s toolbar.  I open the Lertap5.xls workbook, and, after that, use Excel’s Window menu to return to my ScannerEjemploDP2 workbook.  My Excel screen now looks like this:

And that’s it!  Por fin.  I open the Run menu, click on Interpret CCs lines, and, after looking at the Freqs report, return to the Run menu, and click on Elmillon item analysis.  Fat city, mi amor; I’m home free.

Moving towards a turn-key system
The good people at the testing centre can process, I am told, tens of tests, sometimes hundreds, in an academic year.  If the steps I’ve gone through above could be automated, that would be a considerable time saver.  Wouldn’t it be terrific if we could make some Excel macros which would do these things without our intervention, giving us time to attend more meetings, drink more coffee?
Can do?

Yes.

First, consider the matter of making a copy of RawData, renaming the copy as Data, inserting a row at the top, putting in a description of some sort, and then changing the heading on the ID column.  I’ll get Excel to do this by making good use of its macro recorder.
I started as I did above, with a workbook with just two worksheets, RawData and Answers.  With RawData as the active worksheet (the visible one), I clicked on Tools / Macro / Record new macro, and went through the first steps above.  Here’s what Excel created:

Option Explicit

Sub MakeDataSheet()

'

' MakeDataSheet Macro

' Macro recorded 20/05/2005 by nelsonl

'

    Sheets("RawData").Select

    Sheets("RawData").Copy Before:=Sheets(1)

    Sheets("RawData (2)").Select

    Sheets("RawData (2)").Name = "Data"

    Rows("1:1").Select

    Selection.Insert Shift:=xlDown

    ActiveCell.FormulaR1C1 = "Data worksheet for Lertap."

    Range("B2").Select

    ActiveCell.FormulaR1C1 = "ID number"

    Range("A1").Select

End Sub

This part was real easy.  But the CCs part will require more effort.  I’ll start by using the macro recorder again.  First, maybe a thought or two as to my objectives: I’ll get Excel to insert a new worksheet, which I will rename as CCs.  In the first column of the first row of CCs, I’ll put ‘*col (c9-cxx)’; in the first column of the second row I’ll put in ‘*sub title=(’; in the first column of the third row I’ll put in ‘*key’.
Sub MakeCCsSheetPart1()

'

' MakeCCsSheetPart1 Macro

' Macro recorded 20/05/2005 by nelsonl

'

    Sheets("Data").Select

    Sheets.Add

    Sheets("Sheet17").Select

    Sheets("Sheet17").Name = "CCs"

    ActiveCell.FormulaR1C1 = "*col (c9-cxx)"

    Range("A2").Select

    ActiveCell.FormulaR1C1 = "*sub title=("

    Range("A3").Select

    ActiveCell.FormulaR1C1 = "*key"

    Range("A4").Select

End Sub
This little macro has a few problems.  First of all, when I went through the steps, recording as I went, it turned out that the worksheet Excel inserted came to have a name of “Sheet 17”.  I’d bet you twenty copies of Lertap, a year’s supply of green tea, and two cases of Emu Export that the next time I did something like this the name of the inserted workbook would have very little chance of being “Sheet 17”.  I need something more general if I’m going to have a macro that works time after time.

So, have a look at the following code.

The MakeDataSheet subroutine is identical to the code created by Excel’s macro recorder.  The rest of the code was self-made.  A crucial module is the CountItems subroutine, a procedure which determines Nits, the number of test items, and the string of correct answers required by the *key CCs line.  This is quite useful— we’ve now got a routine which can be used with a great variety of tests, each with a different number of items and different item keys.
Another handy bit of code is that seen in Sub CallLertap().  This procedure shows how Lertap may be called into action from within another workbook.

'Start of draft code, 24 May 2005, L Nelson.

Option Explicit

Dim Keys As String

Dim Nits As Integer

Sub MakeDataSheet()

'

' MakeDataSheet Macro

' Macro recorded 20/05/2005 by nelsonl

'

    Sheets("RawData").Select

    Sheets("RawData").Copy Before:=Sheets(1)

    Sheets("RawData (2)").Select

    Sheets("RawData (2)").Name = "Data"

    Rows("1:1").Select

    Selection.Insert Shift:=xlDown

    ActiveCell.FormulaR1C1 = "Data worksheet for Lertap."

    Range("B2").Select

    ActiveCell.FormulaR1C1 = "ID number"

    Range("A1").Select

End Sub

Sub MakeCCsSheet()

    Dim Name As String

    Dim Title As String

      Let Name = Worksheets("Data").Cells(1, 1)

      Let Title = Worksheets("Data").Cells(3, 2)

    Sheets.Add 'Adds a new worksheet.

    Call RenameSheet 'Renames new sheet as CCs

    Call CountItems 'Returns Nits and Keys string.

    Sheets("CCs").Select

      Cells(1, 1) = "*col (c9-c" & Nits + 8 & ")"

      Cells(2, 1) = "*sub title=(" & Title & "), name=(" & Name & ")"

      Cells(3, 1) = "*key " & Keys

      Cells(1, 1).Select

End Sub

Sub RenameSheet()

    Dim Sheet

    For Each Sheet In ActiveWorkbook.Worksheets

      If InStr(1, LCase(Sheet.Name), "sheet") = 1 Then

        'MsgBox "Found sheet named: " & Sheet.Name

        Sheets(Sheet.Name).Name = "CCs"

        Exit For

      End If

    Next Sheet

End Sub

Sub CountItems()

  Dim AnswersRow As Integer

  Dim ItemKey As String

  Let Keys = ""

  Let ItemKey = "X" '(just to get started)

  'The answers are in column 3, from row 2. Get ‘em!

  With Sheets("Answers")

    Let AnswersRow = 1

    Do Until Len(ItemKey) < 1

      Let AnswersRow = AnswersRow + 1

      Let ItemKey = .Cells(AnswersRow, 3)

      Let Keys = Keys & ItemKey

    Loop

    Let Nits = Len(Keys)
    'We now know the number of items, Nits, and _

     we’ve got the keys too.
    'MsgBox "Nits= " & Nits

    'MsgBox "Keys= " & Keys

  End With

End Sub

Sub CallLertap()

  Dim CallLertap5

  On Error GoTo LertapProb

  CallLertap5 = Application.Run("Lertap5.xls!RunCCs")

  Exit Sub

LertapProb:

  On Error GoTo 0

  MsgBox "Uh-oh! Lertap doesn't seem to be open."

End Sub
To set up a “turn-key” system, I’d add the code seen above to the workbook created by the scanner.  Then somewhere I’d add these lines of code:

Call MakeDataSheet

Call MakeCCsSheet

Call CallLertap

These lines could be added to a command button on the RawData worksheet, or, perhaps, to the on-workbook-open event:
Sub Workbook_Open()

  Call MakeDataSheet

  Call MakeCCsSheet

  Call CallLertap

End Sub

If you’re interested in the turn-key idea, you’d probably want to set up your copy of Lertap 5 so that it runs in “production mode”.  In this mode, Lertap can be made to whiz through a whole mess of steps without requiring any user intervention at all.  For example, it’ll go directly from “Interpret CCs lines” to “Elmillon item analysis” without stopping.

To read more about production mode refer to Lelp, the Lertap help system.
If you’ve got Lertap set up in production mode, and if you’ve added code such as that seen above to your scanner-created workbook, then the matter of reformatting the scanner’s output and getting Lertap’s reports and graphs might be extremely simple, perhaps not even requiring a single mouse click (!).

Note: the code above may be freely used— there’s no copyright here.  (Also note that the code is not guaranteed to run correctly.)
Summary

Colleagues at a major Australian testing centre have a scanner / computer duo which will produce an Excel workbook having two worksheets containing test results.  I have called these sheets “RawData” and “Answers”.

The format of these sheets is shown at the start of this document.

In order to get Lertap to process the test results, two worksheets have to be added to the workbook, and formatted to meet Lertap’s criteria.  These worksheets are called Data and CCs.

The Data worksheet can be created using the code seen above in the MakeDataSheet subroutine.  This code was, in fact, made simply by using Excel’s ability to record a macro.

You’ll note that I have assumed that a worksheet description of "Data worksheet for Lertap." is adequate.  This description is placed in the first row of the Data worksheet, and later on it’s picked up by the MakeCCsSheet routine, and then used as the name= assignment on the *col CCs line.

It would be entirely possible to do better here.  We could use an InputBox function to ask for a workbook description, and a suitable title for Lertap, to be entered “on the fly”:

Sub MakeCCsSheet()

    Dim Name As String

    Dim Title As String

  
Let Name = InputBox("Enter a workbook description.")

  
Let Title = InputBox("Enter a title.")
The CCs worksheet is made by the code seen in the MakeCCsSheet code, which calls on RenameSheet and CountItems to get its work done.  The code is flexible, made so that different tests may be processed without having to change the code; each test may have a different number of items.
Finally, to get Lertap to process the test results automatically, the CallLertap routine may be used.  If Lertap is open (its toolbar is showing), and if it’s running in “production mode”, then CallLertap will finish the job for us.  Note that we can do this from within our workbook— we don’t have to actually click on Lertap’s toolbar at any point in this process.
Like these ideas?  Need help?  We’re at: support@lertap5.com.
Setting up a turn-key system, p.11.

